
Description

sts test tests the equality of survivor functions across two or more groups. The log-rank,
Cox, Wilcoxon–Breslow–Gehan, Tarone–Ware, Peto–Peto–Prentice, and Fleming–Harrington tests
are provided, in both unstratified and stratified forms.

sts test also provides a test for trend.

sts test can be used with single- or multiple-record or single- or multiple-failure st data.

Quick start
Log-rank test for the equality of survivor functions across levels of v1 using stset data

sts test v1

Stratified log-rank test for equality of survivor functions across v1 with strata svar

sts test v1, strata(svar)

Same as above, and show tests for each stratum
sts test v1, strata(svar) detail

Log-rank test for equality, and test for a trend in survivor functions for v1
sts test v1, trend

Test equality of survivor functions using the Wilcoxon–Breslow–Gehan test
sts test v1, wilcoxon

Likelihood-ratio test for the equality of survivor functions based on the Cox model
sts test v1, cox

Stratified Cox test of equality of survivor functions with strata svar

sts test v1, cox strata(svar)

Test equality of survivor functions using the Tarone–Ware test
sts test v1, tware

Same as above, and test for a trend using the same weights as used in the Tarone–Ware test
sts test v1, tware trend

Menu
Statistics > Survival analysis > Summary statistics, tests, and tables > Test equality of survivor functions

Syntax

sts test varlist
[

if
] [

in
] [

, options
]

options Description

Main

logrank perform log-rank test of equality; the default
cox perform Cox test of equality
wilcoxon perform Wilcoxon–Breslow–Gehan test of equality
tware perform Tarone–Ware test of equality
peto perform Peto–Peto–Prentice test of equality
fh(p q) perform generalized Fleming–Harrington test of equality
trend test trend of the survivor function across three or more ordered groups
strata(varlist) perform stratified test on varlist, displaying overall test results
detail display individual test results; modifies strata()

Options

mat(mname1 mname2) store vector u in mname1 and matrix V in mname2
noshow do not show st setting information
notitle suppress title

You must stset your data before using sts test; see [ST] stset.
collect is allowed; see [U] 11.1.10 Prefix commands.
Note that fweights, iweights, and pweights may be specified using stset; see [ST] stset.

Options

� � �
Main �

logrank, cox, wilcoxon, tware, peto, and fh(p q) specify the test of equality desired. logrank
is the default, unless the data are pweighted, in which case cox is the default and is the only
possibility.

wilcoxon specifies the Wilcoxon–Breslow–Gehan test; tware, the Tarone–Ware test; peto,
the Peto–Peto–Prentice test; and fh(), the generalized Fleming–Harrington test. The Fleming–
Harrington test requires two arguments, p and q. When p = 0 and q = 0, the Fleming–Harrington
test reduces to the log-rank test; when p = 1 and q = 0, the test reduces to the Mann–Whitney–
Wilcoxon test.

trend specifies that a test for trend of the survivor function across three or more ordered groups be
performed.

strata(varlist) requests that a stratified test be performed.

detail modifies strata(); it requests that, in addition to the overall stratified test, the tests for the
individual strata be reported. detail is not allowed with cox.

� � �
Options �

mat(mname1 mname2) requests that the vector u be stored in mname1 and that matrix V be stored
in mname2. The other tests are rank tests of the form u′V−1u. This option may not be used with
cox.

https://www.stata.com/manuals/u11.pdf#u11.4varnameandvarlists
https://www.stata.com/manuals/u11.pdf#u11.1.3ifexp
https://www.stata.com/manuals/u11.pdf#u11.1.4inrange
https://www.stata.com/manuals/u11.pdf#u11.4varnameandvarlists
https://www.stata.com/manuals/ststset.pdf#ststset
https://www.stata.com/manuals/u11.pdf#u11.1.10Prefixcommands
https://www.stata.com/manuals/ststset.pdf#ststset
https://www.stata.com/manuals/u11.pdf#u11.4varnameandvarlists

noshow prevents sts test from showing the key st variables. This option is seldom used because
most people type stset, show or stset, noshow to set whether they want to see these variables
mentioned at the top of the output of every st command; see [ST] stset.

notitle requests that the title printed above the test be suppressed.

Remarks and examples stata.com

Remarks are presented under the following headings:
The log-rank test
The Wilcoxon (Breslow–Gehan) test
The Tarone–Ware test
The Peto–Peto–Prentice test
The generalized Fleming–Harrington tests
The “Cox” test
The trend test
Video example

sts test tests the equality of the survivor function across groups. With the exception of the Cox
test, these tests are members of a family of statistical tests that are extensions to censored data of
traditional nonparametric rank tests for comparing two or more distributions. A technical description
of these tests can be found in the Methods and formulas section of this entry. Simply, at each distinct
failure time in the data, the contribution to the test statistic is obtained as a weighted standardized
sum of the difference between the observed and expected number of deaths in each of the k groups.
The expected number of deaths is obtained under the null hypothesis of no differences between the
survival experience of the k groups.

The weights or weight function used determines the test statistic. For example, when the weight
is 1 at all failure times, the log-rank test is computed, and when the weight is the number of subjects
at risk of failure at each distinct failure time, the Wilcoxon–Breslow–Gehan test is computed.

The following table summarizes the weights used for each statistical test.

Weight at each distinct
Test failure time (ti)

Log-rank 1

Wilcoxon–Breslow–Gehan ni

Tarone–Ware
√
ni

Peto–Peto–Prentice S̃(ti)

Fleming–Harrington Ŝ(ti−1)
p
{
1− Ŝ(ti−1)

}q
where Ŝ(ti) is the estimated Kaplan–Meier survivor-function value for the combined
sample at failure time ti, S̃(ti) is a modified estimate of the overall survivor function
described in Methods and formulas, and ni is the number of subjects in the risk pool at
failure time ti.

These tests are appropriate for testing the equality of survivor functions across two or more groups.
Up to 800 groups are allowed.

The “Cox” test is related to the log-rank test but is performed as a likelihood-ratio test (or,
alternatively, as a Wald test) on the results from a Cox proportional hazards regression. The log-rank
test should be preferable to what we have labeled the Cox test, but with pweighted data the log-rank
test is not appropriate. Whether you perform the log-rank or Cox test makes little substantive difference
with most datasets.

https://www.stata.com/manuals/ststset.pdf#ststset
http://stata.com


4 sts test — Test equality of survivor functions

sts test, trend can be used to test against the alternative hypothesis that the failure rate
increases or decreases as the level of the k groups increases or decreases. This test is appropriate only
when there is a natural ordering of the comparison groups, for example, when each group represents
an increasing or decreasing level of a therapeutic agent.

trend is not valid when cox is specified.

The log-rank test

sts test, by default, performs the log-rank test, which is, to be clear, the exponential scores test
(Savage 1956; Mantel and Haenszel 1959; Mantel 1963, 1966). This test is most appropriate when
the hazard functions are thought to be proportional across the groups if they are not equal.

This test statistic is constructed by giving equal weights to the contribution of each failure time
to the overall test statistic.

In Testing equality of survivor functions in [ST] sts, we demonstrated the use of this command
with the heart transplant data, a multiple-record, single-failure st dataset.

. use https://www.stata-press.com/data/r18/stan3
(Heart transplant data)

. sts test posttran

Failure _d: died
Analysis time _t: t1

ID variable: id

Equality of survivor functions
Log-rank test

Observed Expected
posttran events events

0 30 31.20
1 45 43.80

Total 75 75.00

chi2(1) = 0.13
Pr>chi2 = 0.7225

We cannot reject the hypothesis that the survivor functions are the same.

sts test, logrank can also perform the stratified log-rank test. Say that it is suggested that
calendar year of acceptance also affects survival and that there are three important periods: 1967–1969,
1970–1972, and 1973–1974. Therefore, a stratified test should be performed:

. stset, noshow

. generate group = 1 if year <= 69
(117 missing values generated)

. replace group=2 if year>=70 & year<=72
(78 real changes made)

. replace group=3 if year>=73
(39 real changes made)

https://www.stata.com/manuals/ststs.pdf#ststsRemarksandexamplesTestingequalityofsurvivorfunctions
https://www.stata.com/manuals/ststs.pdf#ststs

. sts test posttran, strata(group)

Equality of survivor functions
Stratified log-rank test
Strata: group

Observed Expected
posttran events events*

0 30 31.51
1 45 43.49

Total 75 75.00

* aggregated over stratum-specific results

chi2(1) = 0.20
Pr>chi2 = 0.6547

When a stratified test is requested, Stata computes the expected events under the null hypothesis
separately for each stratum, and those results are aggregated over strata.

We now want to show the within-stratum tests, so we type

. sts test posttran, strata(group) detail

Equality of survivor functions
Stratified log-rank test
Strata: group

-> group = 1

Observed Expected
posttran events events

0 14 13.59
1 17 17.41

Total 31 31.00

chi2(1) = 0.03
Pr>chi2 = 0.8558

-> group = 2

Observed Expected
posttran events events

0 13 13.63
1 20 19.37

Total 33 33.00

chi2(1) = 0.09
Pr>chi2 = 0.7663

-> group = 3

Observed Expected
posttran events events

0 3 4.29
1 8 6.71

Total 11 11.00

chi2(1) = 0.91
Pr>chi2 = 0.3410

-> Total

Observed Expected
posttran events events*

0 30 31.51
1 45 43.49

Total 75 75.00

* aggregated over stratum-specific results

chi2(1) = 0.20
Pr>chi2 = 0.6547

The Wilcoxon (Breslow–Gehan) test

sts test, wilcoxon performs the generalized Wilcoxon test of Breslow (1970) and Gehan (1965).
This test is appropriate when hazard functions are thought to vary in ways other than proportionally
and when censoring patterns are similar across groups.

The Wilcoxon test statistic is constructed by weighting the contribution of each failure time to
the overall test statistic by the number of subjects at risk. Thus it gives heavier weights to earlier
failure times when the number at risk is higher. As a result, this test is susceptible to differences in
the censoring pattern of the groups.

sts test, wilcoxon works the same way as sts test, logrank:

. sts test posttran, wilcoxon

Equality of survivor functions
Wilcoxon--Breslow--Gehan test

Observed Expected Sum of
posttran events events ranks

0 30 31.20 -85
1 45 43.80 85

Total 75 75.00 0

chi2(1) = 0.14
Pr>chi2 = 0.7083

With the strata() option, sts test, wilcoxon performs the stratified test:

. sts test posttran, wilcoxon strata(group)

Equality of survivor functions
Stratified Wilcoxon--Breslow--Gehan test
Strata: group

Observed Expected Sum of
posttran events events* ranks*

0 30 31.51 -40
1 45 43.49 40

Total 75 75.00 0

* aggregated over stratum-specific results

chi2(1) = 0.22
Pr>chi2 = 0.6385

As with sts test, logrank, you can also specify the detail option to see the within-stratum tests.

The Tarone–Ware test
sts test, tware performs a test suggested by Tarone and Ware (1977), with weights equal to

the square root of the number of subjects in the risk pool at time ti.

Like Wilcoxon’s test, this test is appropriate when hazard functions are thought to vary in ways
other than proportionally and when censoring patterns are similar across groups. The test statistic
is constructed by weighting the contribution of each failure time to the overall test statistic by the
square root of the number of subjects at risk. Thus, like the Wilcoxon test, it gives heavier weights,
although not as large, to earlier failure times. Although less susceptible to the failure and censoring
pattern in the data than Wilcoxon’s test, this could remain a problem if large differences in these
patterns exist between groups.

sts test, tware works the same way as sts test, logrank:

. sts test posttran, tware

Equality of survivor functions
Tarone--Ware test

Observed Expected Sum of
posttran events events ranks

0 30 31.20 -9.3375685
1 45 43.80 9.3375685

Total 75 75.00 0

chi2(1) = 0.12
Pr>chi2 = 0.7293

With the strata() option, sts test, tware performs the stratified test:

. sts test posttran, tware strata(group)

Equality of survivor functions
Stratified Tarone--Ware test
Strata: group

Observed Expected Sum of
posttran events events* ranks*

0 30 31.51 -7.4679345
1 45 43.49 7.4679345

Total 75 75.00 0

* aggregated over stratum-specific results

chi2(1) = 0.21
Pr>chi2 = 0.6464

As with sts test, logrank, you can also specify the detail option to see the within-stratum tests.

The Peto–Peto–Prentice test
sts test, peto performs an alternative to the Wilcoxon test proposed by Peto and Peto (1972)

and Prentice (1978). The test uses as the weight function an estimate of the overall survivor function,
which is similar to that obtained using the Kaplan–Meier estimator. See Methods and formulas for
details.

This test is appropriate when hazard functions are thought to vary in ways other than proportionally,
but unlike the Wilcoxon–Breslow–Gehan test, it is not affected by differences in censoring patterns
across groups.

sts test, peto works the same way as sts test, logrank:

. sts test posttran, peto

Equality of survivor functions
Peto--Peto--Prentice test

Observed Expected Sum of
posttran events events ranks

0 30 31.20 -.86708453
1 45 43.80 .86708453

Total 75 75.00 0

chi2(1) = 0.15
Pr>chi2 = 0.6979

With the strata() option, sts test, peto performs the stratified test:

. sts test posttran, peto strata(group)

Equality of survivor functions
Stratified Peto--Peto--Prentice test
Strata: group

Observed Expected Sum of
posttran events events* ranks*

0 30 31.51 -1.4212233
1 45 43.49 1.4212233

Total 75 75.00 0

* aggregated over stratum-specific results

chi2(1) = 0.43
Pr>chi2 = 0.5129

As with the previous tests, you can also specify the detail option to see the within-stratum tests.

The generalized Fleming–Harrington tests

sts test, fh(p q) performs the Harrington and Fleming (1982) class of test statistics. The
weight function at each distinct failure time, t, is the product of the Kaplan–Meier survivor estimate
at time t− 1 raised to the p power and 1 − the Kaplan–Meier survivor estimate at time t− 1 raised
to the q power. Thus, when specifying the Fleming and Harrington option, you must specify two
nonnegative arguments, p and q.

When p > q, the test gives more weights to earlier failures than to later ones. When p < q, the
opposite is true, and more weight is given to later failures than to earlier ones. When p and q are
both zero, the weight is 1 at all failure times and the test reduces to the log-rank test.

sts test, fh(p q) works the same way as sts test, logrank. If we specify p = 0 and q = 0
we will get the same results as the log-rank test:

. sts test posttran, fh(0 0)

Equality of survivor functions
Fleming--Harrington test
Powers: p = 0, q = 0

Observed Expected Sum of
posttran events events ranks

0 30 31.20 -1.1995511
1 45 43.80 1.1995511

Total 75 75.00 0

chi2(1) = 0.13
Pr>chi2 = 0.7225

We could, for example, give more weight to later failures than to earlier ones.

. sts test posttran, fh(0 3)

Equality of survivor functions
Fleming--Harrington test
Powers: p = 0, q = 3

Observed Expected Sum of
posttran events events ranks

0 30 31.20 -.10288411
1 45 43.80 .10288411

Total 75 75.00 0

chi2(1) = 0.01
Pr>chi2 = 0.9065

Similarly to the previous tests, with the strata() option, sts test, fh() performs the stratified
test:

. sts test posttran, fh(0 3) strata(group)

Equality of survivor functions
Stratified Fleming--Harrington test
Powers: p = 0, q = 3
Strata: group

Observed Expected Sum of
posttran events events* ranks*

0 30 31.51 -.05315105
1 45 43.49 .05315105

Total 75 75.00 0

* aggregated over stratum-specific results

chi2(1) = 0.00
Pr>chi2 = 0.9494

As with the other tests, you can also specify the detail option to see the within-stratum tests.

The “Cox” test
The term Cox test is our own, and this test is a variation on the log-rank test using Cox regression.

One way of thinking about the log-rank test is as a Cox proportional hazards model on indicator
variables for each of the groups. The log-rank test is a test that the coefficients are zero or, if you
prefer, that the hazard ratios are one. The log-rank test is, in fact, a score test of that hypothesis
performed on a slightly different (partial) likelihood function that handles ties more accurately.

Many researchers think that a (less precise) score test on the precise likelihood function is preferable
to a (more precise) likelihood-ratio test on the approximate likelihood function used in Cox regression
estimation. In our experience, it makes little difference:

. sts test posttran, cox

Equality of survivor functions
Cox regression-based test

Observed Expected Relative
posttran events events hazard

0 30 31.20 0.9401
1 45 43.80 1.0450

Total 75 75.00 1.0000

LR chi2(1) = 0.13
Pr>chi2 = 0.7222

Note: Reporting LR model test from Cox regression
on posttran.

By comparison, sts test, logrank also reported χ2 = 0.13, although the significance level was
0.7225, meaning that the χ2 values differed in the fourth digit. As mentioned by Kalbfleisch and
Prentice (2002, 20), a primary advantage of the log-rank test is the ease with which it can be
explained to nonstatisticians, because the test statistic is the difference between the observed and
expected number of failures within groups.

Our purpose in offering sts test, cox is not to promote its use instead of the log-rank test but
to provide a test for researchers with sample-weighted data.

If you have sample weights (if you specified pweights when you stset the data), you cannot
run the log-rank or Wilcoxon tests. The Cox regression model, however, has been generalized to
sample-weighted data, and Stata’s stcox can fit models with such data. In sample-weighted data, the
likelihood-ratio statistic is no longer appropriate, but the Wald test based on the robust estimator of
variance is.

Thus if we treated these data as sample-weighted data, we would obtain

. generate one = 1

. stset t1 [pw=one], id(id) time0(_t0) failure(died) noshow

Survival-time data settings

ID variable: id
Failure event: died!=0 & died<.

Observed time interval: (_t0, t1]
Exit on or before: failure

Weight: [pweight=one]

172 total observations
0 exclusions

172 observations remaining, representing
103 subjects
75 failures in single-failure-per-subject data

31,938.1 total analysis time at risk and under observation
At risk from t = 0

Earliest observed entry t = 0
Last observed exit t = 1,799

. sts test posttran, cox

Equality of survivor functions
Cox regression-based test

Observed Expected Relative
posttran events events hazard

0 30.00 31.20 0.9401
1 45.00 43.80 1.0450

Total 75.00 75.00 1.0000

Wald chi2(1) = 0.13
Pr>chi2 = 0.7181

Note: Reporting Wald model test from Cox
regression on posttran.

sts test, cox now reports the Wald statistic, which is, to two digits, 0.13, just like all the others.

The trend test
When the groups to be compared have a natural order, such as increasing or decreasing age groups

or drug dosage, you may want to test the null hypothesis that there is no difference in failure rate
among the groups versus the alternative hypothesis that the failure rate increases or decreases as you
move from one group to the next.

We illustrate this test with a dataset from a carcinogenesis experiment reprinted in Marubini and
Valsecchi (1995, 126). Twenty-nine experimental animals were exposed to three levels (0, 1.5, 2.0)
of a carcinogenic agent. The time in days to tumor formation was recorded. Here are a few of the
observations:

. use https://www.stata-press.com/data/r18/marubini, clear

. list time event group dose in 1/9

time event group dose

1. 67 1 2 1.5
2. 150 1 2 1.5
3. 47 1 3 2
4. 75 0 1 0
5. 58 1 3 2

6. 136 1 2 1.5
7. 58 1 3 2
8. 150 1 2 1.5
9. 43 0 2 1.5

In these data, there are two variables that indicate exposure level. The group variable is coded 1,
2, and 3, indicating a one-unit separation between exposures. The dose variable records the actual
exposure dosage. To test the null hypothesis of no difference among the survival experience of the
three groups versus the alternative hypothesis that the survival experience of at least one of the groups
is different, it does not matter if we use group or dose.

. stset time, fail(event) noshow

Survival-time data settings

Failure event: event!=0 & event<.
Observed time interval: (0, time]

Exit on or before: failure

29 total observations
0 exclusions

29 observations remaining, representing
15 failures in single-record/single-failure data

2,564 total analysis time at risk and under observation
At risk from t = 0

Earliest observed entry t = 0
Last observed exit t = 246

. sts test group

Equality of survivor functions
Log-rank test

Observed Expected
group events events

1 4 6.41
2 6 6.80
3 5 1.79

Total 15 15.00

chi2(2) = 8.05
Pr>chi2 = 0.0179

. sts test dose

Equality of survivor functions
Log-rank test

Observed Expected
dose events events

0 4 6.41
1.5 6 6.80

2 5 1.79

Total 15 15.00

chi2(2) = 8.05
Pr>chi2 = 0.0179

For the trend test, however, the distance between the values is important, so using group or dose
will produce different results.

. sts test group, trend

Equality of survivor functions
Log-rank test

Observed Expected
group events events

1 4 6.41
2 6 6.80
3 5 1.79

Total 15 15.00

chi2(2) = 8.05
Pr>chi2 = 0.0179

Test for trend in survivor functions

chi2(1) = 5.87
Pr>chi2 = 0.0154

. sts test dose, trend

Equality of survivor functions
Log-rank test

Observed Expected
dose events events

0 4 6.41
1.5 6 6.80

2 5 1.79

Total 15 15.00

chi2(2) = 8.05
Pr>chi2 = 0.0179

Test for trend in survivor functions

chi2(1) = 3.66
Pr>chi2 = 0.0557

Although the above trend test was constructed using the log-rank test, any of the previously
mentioned weight functions can be used. For example, a trend test on the data can be performed
using the same weights as the Peto–Peto–Prentice test by specifying the peto option.

. sts test dose, trend peto

Equality of survivor functions
Peto--Peto--Prentice test

Observed Expected Sum of
dose events events ranks

0 4 6.41 -1.2792221
1.5 6 6.80 -1.3150418

2 5 1.79 2.5942639

Total 15 15.00 0

chi2(2) = 8.39
Pr>chi2 = 0.0150

Test for trend in survivor functions

chi2(1) = 2.85
Pr>chi2 = 0.0914

Video example

How to test the equality of survivor functions using nonparametric tests

Stored results
sts test stores the following in r():

Scalars
r(df) degrees of freedom r(chi2) χ2

r(df tr) degrees of freedom, trend test r(chi2 tr) χ2, trend test

Methods and formulas
Let t1 < t2 < · · · < tk denote the ordered failure times; let dj be the number of failures at tj

and nj be the population at risk just before tj ; and let dij and nij denote the same things for group
i, i = 1, . . . , r.

We are interested in testing the null hypothesis

H0 : λ1(t) = λ2(t) = · · · = λr(t)

where λ(t) is the hazard function at time t, against the alternative hypothesis that at least one of the
λi(t) is different for some tj .

As described in Klein and Moeschberger (2003, 205–216), Kalbfleisch and Prentice (2002, 20–22),
and Collett (2015, 50–51), if the null hypothesis is true, the expected number of failures in group i
at time tj is eij = nijdj/nj , and the test statistic

u′ =

k∑
j=1

W (tj)(d1j − e1j , . . . , drj − erj)

https://www.youtube.com/watch?v=W1uympJV7Ko&list=UUVk4G4nEtBS4tLOyHqustDA

is formed. W (tj) is a positive weight function defined as zero when nij is zero. The various test
statistics are obtained by selecting different weight functions, W (tj). See the table in the Remarks
and examples section of this entry for a list of these weight functions. For the Peto–Peto–Prentice
test,

W (tj) = S̃(tj) =
∏

`:t`≤tj

(
1− d`

n` + 1

)

The variance matrix V for u has elements

Vil =

k∑
j=1

W (tj)
2nijdj(nj − dj)
nj(nj − 1)

(
δil −

nij
nj

)
where δil = 1 if i = l and δil = 0, otherwise.

For the unstratified test, statistic u′V−1u is distributed as χ2 with r − 1 degrees of freedom.

For the stratified test, let us and Vs be the results of performing the above calculation separately
within stratum, and define u =

∑
s us and V =

∑
sVs. The χ2 test is given by u′V−1u redefined

in this way.

The “Cox” test is performed by fitting a (possibly stratified) Cox regression using stcox on r− 1
indicator variables, one for each group with one of the indicators omitted. The χ2 test reported is then
the likelihood-ratio test (no pweights) or the Wald test (based on the robust estimate of variance);
see [ST] stcox.

The reported relative hazards are the exponentiated coefficients from the Cox regression renor-
malized, and the renormalization plays no role in calculating the test statistic. The renormalization
is chosen so that the expected-number-of-failures-within-group weighted average of the regression
coefficients is 0 (meaning that the hazard is 1). Let bi, i = 1, . . . , r−1, be the estimated coefficients,
and define br = 0. The constant K is then calculated with

K =

r∑
i=1

eibi/d

where ei =
∑
j eij is the expected number of failures for group i, d is the total number of failures

across all groups, and r is the number of groups. The reported relative hazards are exp(bi −K).

The trend test assumes that there is natural ordering of the r groups, r > 2. Here we are interested
in testing the null hypothesis

H0 : λ1(t) = λ2(t) = · · · = λr(t)

against the alternative hypothesis

Ha : λ1(t) ≤ λ2(t) ≤ · · · ≤ λr(t)

https://www.stata.com/manuals/ststcox.pdf#ststcox

The test uses u as previously defined with any of the available weight functions. The test statistic
is given by

(∑r
i=1 aiui

)2

a′Va

where a1 ≤ a2 ≤ · · · ≤ ar are scores defining the relationship of interest. A score is assigned to
each comparison group, equal to the value of the grouping variable for that group. a is the vector of
these scores.

Also see
[ST] stcox — Cox proportional hazards model

[ST] sts — Generate, graph, list, and test the survivor and related functions

[ST] sts generate — Create variables containing survivor and related functions

[ST] sts graph — Graph the survivor or related function

[ST] sts list — List the survivor or related function

[ST] stset — Declare data to be survival-time data

[PSS-2] power exponential — Power analysis for a two-sample exponential test

[PSS-2] power logrank — Power analysis for the log-rank test

